ALGORITHM 98

EVALUATION OF DEFINITE COMPLEX LINE INTEGRALS

JOHN L. PFALTZ

Syracuse University Computing Center, Syracuse, N. Y.

```
procedure COMPLINEINTGRL(A, B, N, RSSUM);
  value A, B, N; real A, B, N; array RSSUM;
comment COMPLINEINTGRL approximates the complex line
  integral by evaluating the partial Riemann-Stieltjes sum
  Σ_{t=1}^{n} f(z_k)[z_t − z_{t−1}] where a ≦ t ≦ b and z_k ∈ (z_{t−1}, z_t). The
  programmer must provide 1) the procedures GAMMA(T, Z) to
  calculate z(t) on Γ, and FUNCT(Z, F) to calculate function
  values, and 2) the end points A and B of the parametric interval
  and N the number of subintervals into which [a, b] is to be
  partitioned;
begin integer I; real T, DELT; real array ZT, ZTL, DELZ,
    ZK, PART[1:2]; RSSUM[1] := 0.0; RSSUM[2] := 0.0;
    DELT := (B − A)/N; T := A;
line: GAMMA(T, ZT);
    if T = A then go to next;
    for I := 1 step 1 until 2 do
    begin
      DELZ[I] := ZT[I] − ZTL[I]; end;
    for I := 1 step 1 until 2 do
    begin
      ZK[I] := ZTL[I] + DELZ[I]/2.0; end;
    FUNCT(ZK, FZ);
    PART[1] := FZ[1] × DELZ[1] − FZ[2] × DELZ[2];
    PART[2] := FZ[1] × DELZ[2] + FZ[2] × DELZ[1];
    for I := 1 step 1 until 2 do
    begin
      RSSUM[I] := RSSUM[I] + PART[I]; end;
    if T < B − (0.25 × DELT) then go to next else go to
      exit;
next: for I := 1 step 1 until 2 do
    begin
      ZTL[I] := ZT[I]; end;
    T := T + DELT;
    go to line;
exit: end COMPLINEINTGRL.
```